

# REDUCTION OF MISORIENTATIONS BETWEEN TWO CUBIC CRYSTALS INTO THE BASE DOMAIN OF AXIS-ANGLE SPACE

K. SZTWIERTNIA and J. POSPIECH

*Instytut Podstaw Metalurgii, Krakow, Poland.*

F. HAESSNER

*Institut für Werkstoffe, Technische Universität Braunschweig, Germany*

*(Received January 20, 1990)*

A misorientation between two cubic crystallites can be described using different orientation parameters and represented in many symmetrically equivalent ways. It is, however, convenient when the misorientations between different pairs of crystallites all lie in the same base domain. This paper presents a computer program which reduces a misorientation given in axis-angle or in Euler's space into the base domain of the smallest angle of rotation.

KEY WORDS Misorientation, cubic symmetry, axis-angle space.

## INTRODUCTION

In many investigations of polycrystalline materials the misorientations between adjacent crystallites play an important role. A particular misorientation can be represented in many symmetrically equivalent ways. The number of different possibilities depends on the lattice symmetry of the material.

For a single misorientation it is immaterial which of the symmetrically equivalent describing modes is used. In an individual case one would choose out of the manifold possibilities the description that most clearly represents the case in question. The situation is quite a different one when two or more misorientations are to be compared. Frequency analyses of misorientations are examples of this. Here the respective misorientations must be made mutually compatible. A reduction of this kind is possible. It was first time introduced for the case of cubic symmetry by Mackenzie (1958).

Although the defined reduction of the multiplicity of describing modes has long been known, the implementation requires great effort. We have therefore developed a computer programme (in FORTRAN 77) for the reduction of an arbitrary misorientation which is suitable for PC computers. Before describing the programme, to better understand the problem of multiplicity it will be analytically discussed in a separate section.

## THE MISORIENTATION BETWEEN TWO CUBIC CRYSTALLITES

Let  $A$  and  $B$  be two crystallites of the same material, the coordinate systems  $K_A$  and  $K_B$  of which are parallel to  $\langle 100 \rangle$  directions in the respective lattices. The misorientation  $\Gamma_{AB}$  is defined as the rotation which transforms the reference system  $K_B$  into  $K_A$ :

$$K_A = \Gamma_{AB} \cdot K_B = \Gamma \cdot K_B. \quad (1)$$

Since both crystallites considered have cubic symmetry there are many different ways in which the misorientation between them can be described. The symmetrically equivalent descriptions  $\Gamma_{AB}^e$  result when the symmetry elements of the O-group are taken into account. These are: (a) The identity element, (b) rotations of  $180^\circ$  around the three  $\langle 100 \rangle$  axes, (c) rotations of  $90^\circ$  around the same axes, (d) rotations of  $180^\circ$  around the six  $\langle 110 \rangle$  axes, (e) rotations of  $120^\circ$  around the four  $\langle 111 \rangle$  axes. The orthogonal matrices  $S_i$  ( $i = 1, \dots, 24$ ) corresponding to these 24 rotations are collated in Table 1.

The symmetrically equivalent descriptions of the misorientation  $\Gamma_{AB}$  are then given by:

$$\Gamma_{AB}^e = S_i \cdot \Gamma \cdot S_j. \quad (2)$$

$(i, j = 1, \dots, 24)$

Since the two crystallites are indistinguishable, the inverse rotation  $\Gamma_{BA}$  is equivalent to  $\Gamma_{AB}$ . This means that  $\Gamma$  and  $\Gamma^{-1} = \Gamma^T$  (since  $\Gamma$  is real orthogonal) are equivalent:

$$\Gamma_{BA}^e = S_i \cdot \Gamma^{-1} \cdot S_j. \quad (3)$$

In consequence, there are  $24 \times 24 \times 2 = 1152$  equally valid descriptions of the misorientation.

A given misorientation can be described using different orientation parameters, the choice of which depends on the nature of the question to be answered. In the present case the misorientation is described in the axis-angle space by a rotation axis  $\mathbf{r}$  and an angle of rotation  $\omega$  which are common to both crystallites (Hansen *et al.*, 1978). By rotation around the axis the coordinate system of the first crystallite is transferred into that of the other. The rotation axis can be specified by a unit vector  $\mathbf{r}$  with components  $r_x, r_y, r_z$  or by using spherical polar coordinates  $\vartheta, \psi$ . Hence the misorientation  $\Gamma$  can be described as the rotation  $(\omega, \mathbf{r})$  or as the rotation  $(\omega, \vartheta, \psi)$ . In matrix representation it follows:

$$\Gamma = \begin{pmatrix} (1 - r_x^2) \cdot \cos \omega + r_x^2 & r_x \cdot r_y \cdot c + r_z \cdot \sin \omega & r_x \cdot r_z \cdot c - r_y \cdot \sin \omega \\ r_x \cdot r_y \cdot c - r_z \cdot \sin \omega & (1 - r_y^2) \cdot \cos \omega + r_y^2 & r_y \cdot r_z \cdot c + r_x \cdot \sin \omega \\ r_x \cdot r_z \cdot c + r_y \cdot \sin \omega & r_y \cdot r_z \cdot c - r_x \cdot \sin \omega & (1 - r_z^2) \cdot \cos \omega + r_z^2 \end{pmatrix} \quad (4)$$

with  $r_x = \cos \psi \cdot \sin \vartheta$ ,  $r_y = \sin \psi \cdot \sin \vartheta$ ,  $r_z = \cos \vartheta$ ,  $c = 1 - \cos \omega$ .

For materials without crystal symmetry it is sufficient to describe all the physically distinguishable misorientations in the whole axis-angle space by the sub-domain defined by the interval  $0 \leq \omega \leq \pi$ ,  $0 \leq \vartheta \leq \pi$ ,  $0 \leq \psi \leq 2\pi$ . Where crystal symmetry prevails, additional conditions result which restrict the size of the sub-domain further, as will be described in the following. It is practicable to choose the base domain so that within it the absolute value of the angle of

**Table 1** The 24 possible symmetry elements of the O-group.

<i>Symbol</i>	<i>Matrix</i>	<i>Symbol</i>	<i>Matrix</i>	<i>Symbol</i>	<i>Matrix</i>
$S_1 = E$	$\begin{matrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{matrix}$	$S_9 = L_{\bar{1}\bar{1}\bar{1}}^3$	$\begin{matrix} 0 & 0 & 1 \\ 1 & 0 & 0 \\ 0 & 1 & 0 \end{matrix}$	$S_{17} = L_{01\bar{1}}^2$	$\begin{matrix} -1 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & -1 & 0 \end{matrix}$
$S_2 = L_{010}^2$	$\begin{matrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{matrix}$	$S_{10} = L_{1\bar{1}\bar{1}}^3$	$\begin{matrix} 0 & 0 & -1 \\ 1 & 0 & 0 \\ 0 & -1 & 0 \end{matrix}$	$S_{18} = L_{100}^4$	$\begin{matrix} 1 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & 1 & 0 \end{matrix}$
$S_3 = L_{001}^2$	$\begin{matrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{matrix}$	$S_{11} = L_{11\bar{1}}^3$	$\begin{matrix} 0 & 0 & -1 \\ -1 & 0 & 0 \\ 0 & 1 & 0 \end{matrix}$	$S_{19} = L_{100}^4$	$\begin{matrix} 1 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & -1 & 0 \end{matrix}$
$S_4 = L_{100}^2$	$\begin{matrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{matrix}$	$S_{12} = L_{\bar{1}\bar{1}1}^3$	$\begin{matrix} 0 & 0 & 1 \\ -1 & 0 & 0 \\ 0 & -1 & 0 \end{matrix}$	$S_{20} = L_{011}^2$	$\begin{matrix} -1 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 1 & 0 \end{matrix}$
$S_5 = L_{111}^3$	$\begin{matrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 1 & 0 & 0 \end{matrix}$	$S_{13} = L_{210}^2$	$\begin{matrix} 0 & 0 & -1 \\ 0 & -1 & 0 \\ -1 & 0 & 0 \end{matrix}$	$S_{21} = L_{110}^2$	$\begin{matrix} 0 & -1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & -1 \end{matrix}$
$S_6 = L_{1\bar{1}\bar{1}}^3$	$\begin{matrix} 0 & -1 & 0 \\ 0 & 0 & 1 \\ -1 & 0 & 0 \end{matrix}$	$S_{14} = L_{10\bar{1}}^2$	$\begin{matrix} 0 & 0 & 1 \\ 0 & -1 & 0 \\ 1 & 0 & 0 \end{matrix}$	$S_{22} = L_{001}^4$	$\begin{matrix} 0 & 1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & 1 \end{matrix}$
$S_7 = L_{\bar{1}\bar{1}\bar{1}}^3$	$\begin{matrix} 0 & -1 & 0 \\ 0 & 0 & -1 \\ 1 & 0 & 0 \end{matrix}$	$S_{15} = L_{010}^4$	$\begin{matrix} 0 & 0 & 1 \\ 0 & 1 & 0 \\ -1 & 0 & 0 \end{matrix}$	$S_{23} = L_{110}^2$	$\begin{matrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & -1 \end{matrix}$
$S_8 = L_{\bar{1}\bar{1}1}^3$	$\begin{matrix} 0 & 1 & 0 \\ 0 & 0 & -1 \\ -1 & 0 & 0 \end{matrix}$	$S_{16} = L_{010}^4$	$\begin{matrix} 0 & 0 & -1 \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{matrix}$	$S_{24} = L_{00\bar{1}}^4$	$\begin{matrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{matrix}$

The symbols  $L_{uvw}^n$  represents the matrix for a  $n$ -fold rotation of the crystal coordinate system around the  $uvw$  crystal direction (after Hansen *et al.*, 1978).

rotation  $\omega$  is the smallest of all possible angles of rotation  $\omega_j$ . In the case of cubic symmetry 24 different angles of rotation  $\omega_j$  exist. They may be obtained using the following expression:

$$\omega_j = \arccos ((\text{Tr} (\Gamma \cdot S_j) - 1)/2). \tag{5}$$

These results are a consequence of the following facts:

(a)  $\text{Tr} (\Gamma) = \text{Tr} (\Gamma^{-1}) = 1 + 2 \cos \omega$ .

(b) for every  $C$  and  $D$  it holds that  $\text{Tr} (CD) = \text{Tr} (DC)$ , on condition that both products exist,

(c) the product  $S_k \cdot S_i$  is also a symmetry rotation of the O-group.

The smallest angle of rotation  $\omega_d$  according to (5), i.e.

$$\omega_d = \arccos \left( \left( \text{Max}_j \text{Tr} (\Gamma \cdot S_j) - 1 \right) / 2 \right) \tag{6}$$

shall be denoted ‘‘angle of disorientation.’’ Where both crystallites exhibit cubic

symmetry, the largest value the angle of disorientation may have is  $\sim 62.8^\circ$ , as first shown by Mackenzie (1958).

Further subdivision of the  $\omega = \text{const.}$  region may be concluded from the following: If  $\Gamma$  represents a rotation  $\omega$ ,  $r$ , then  $S_i \cdot \Gamma \cdot S_i^{-1}$  represents a rotation  $\omega$ ,  $S_i \cdot r$ . The interval for the rotation angle was defined as  $0 \leq \omega \leq \pi$ . Then, if the inversion is not taken into account, the position of the rotation axis can always be reduced to a double standard triangle. Through inversion a further equality results:  $(\omega, \vartheta, \psi)^e = (\omega, \vartheta, \pi/2 - \psi)$ .<sup>†</sup> It follows from this that the axis of rotation lies in the single spherical standard triangle, determined f.i. by  $r_z \geq r_x \geq r_y \geq 0$ .

### References

1. Mackenzie, J. K. (1958). *Biometrika* **45**, 229.
2. Hansen, J., Pospiech, J., Lücke, K. (1978). *Tables for Texture Analysis of Cubic Crystals*, Springer Verlag, Berlin-Heidelberg.

---

<sup>†</sup> The equality  $(\omega, \vartheta, \psi)^e = (\omega, \vartheta, \pi/2 - \psi)$  can be obtained from the relation  $\Gamma^e = S_{21} \cdot \Gamma^{-1} \cdot S_{21}^{-1}$ .

## THE COMPUTER PROGRAM

C This program reduces the given misorientation to the base domain of  
 C the disorientation angle. The rotation axis lies in the standard  
 C triangle. The program prints the disorientation angle and the para-  
 C meters of the rotation axis. The rotation axis is represented by  
 C polar coordinates theta, psi and additionally by components x,y,z.  
 C

```

COMMON /SMA/ W(3,3)
COMMON /SMBA/ XX,RXM
XX=.0174533
RXM=45.*XX
1 PRINT *, 'The misorientation described by rotational coordinates -
*1 - enter, the misorientation described by the Euler
*Angles - 2 - enter.'
READ *, TYP
2 CONTINUE
RX=10000.
IF(TYP.EQ.1.) THEN
PRINT *, 'Give the rotation angle and the coordinates of rotation
*axis - omega, x, y, z,'
READ *, OM,X,Y,Z
CALL MATR1(OM,X,Y,Z)
END IF
IF(TYP.EQ.2.) THEN
PRINT *, 'Give the Euler Angles - fi1, fi, fi2'
READ *, FI1,FI,FI2
CALL MATR2(FI1,FI,FI2)
END IF
IF(TYP.NE.1..AND.TYP.NE.2.) GO TO 1
CALL MINOS(KR,RX)
IF(RX.EQ.0.) THEN
PRINT *, 'omega=0'
GO TO 3
END IF
CALL AXIS(KR,RX,A1,A2,A3)
CALL BASIS(A1,A2,A3,PR,TH,B1,B2,B3)
RX=RX/XX
IF(PR.GT.45.) PR=90.-PR
PRINT 100, RX,TH,PR,B1,B2,B3
3 PRINT *, 'For continue 1 - enter'
READ *, DES
IF(DES.EQ.1.) GO TO 2
100 FORMAT(1H ,7H omega=,F5.1,8H theta=,f5.1,6H psi=,f5.1,3H***,
*3F7.3)
STOP
END

SUBROUTINE MATR1(OM,X,Y,Z)
COMMON /SMA/ W(3,3)
COMMON /SMBA/ XX,RXM
OM=OM*XX
SQ=SQRT(X*X+Y*Y+Z*Z)
X=X/SQ
Y=Y/SQ
Z=Z/SQ
SM=SIN(OM)
CM=COS(OM)
COM=1.-CM
W(1,1)=(1.-X*X)*CM+X*X
W(1,2)=X*Y*COM+Z*SM
W(1,3)=X*Z*COM-Y*SM
W(2,1)=X*Y*COM-Z*SM

```

```

W(2,2)=(1.-Y*Y)*CM+Y*Y
W(2,3)=Y*Z*COM+X*SM
W(3,1)=X*Z*COM+Y*SM
W(3,2)=Y*Z*COM-X*SM
W(3,3)=(1.-Z*Z)*CM+Z*Z
RETURN
END

```

```

SUBROUTINE MATR2(F1,FD,F2)
COMMON /SMA/ W(3,3)
COMMON /SMBA/ XX,RXM
F1=F1*XX
FD=FD*XX
F2=F2*XX
SA1=SIN(F1)
SAD=SIN(FD)
SA2=SIN(F2)
CA1=COS(F1)
CAD=COS(FD)
CA2=COS(F2)
W(1,1)=CA1*CA2-SA1*SA2*CAD
W(1,2)=SA1*CA2+CA1*SA2*CAD
W(1,3)=SA2*SAD
W(2,1)=-CA1*SA2-SA1*CA2*CAD
W(2,2)=-SA1*SA2+CA1*CA2*CAD
W(2,3)=CA2*SAD
W(3,1)=SA1*SAD
W(3,2)=-CA1*SAD
W(3,3)=CAD
RETURN
END

```

```

SUBROUTINE MINOS(KR,RX)
COMMON/SMA/ W(3,3)
COMMON/SMBA/ XX,RXM
N=0
W11=W(1,1)
W12=W(1,2)
W13=W(1,3)
W21=W(2,1)
W22=W(2,2)
W23=W(2,3)
W31=W(3,1)
W32=W(3,2)
W33=W(3,3)
COM=ZZZ(W11,W22,W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W11,W22,-W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W11,-W22,W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W11,-W22,-W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W21,W32,W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W21,W32,-W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W21,-W32,W13)

```

```

CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W21,-W32,-W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W31,W12,W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W31,W12,-W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W31,-W12,W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W31,-W12,-W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W31,-W22,-W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W31,-W22,W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W31,W22,-W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W31,W22,W13)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W11,-W32,-W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W11,-W32,W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W11,W32,-W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W11,W32,W23)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W21,-W12,-W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W21,-W12,W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(W21,W12,-W33)
CALL MINOM(KR,RX,COM,N)
IF(RX.LT.RXM) RETURN
COM=ZZZ(-W21,W12,W33)
CALL MINOM(KR,RX,COM,N)
RETURN
END

SUBROUTINE MINOM(KR,RX,COM,N)
N=N+1
IF(COM.GT.RX) RETURN
RX=COM
KR=N
RETURN
END

```

```

FUNCTION ZZZ(A,B,C)
CCC=(A+B+C)/2.0-0.5
IF(CCC.GT.1.) CCC=1.
IF(CCC.LT.-1.) CCC=-1.
ZZZ=ACOS(CCC)
RETURN
END

```

```

SUBROUTINE BASIS(D1,D2,D3,PR,TH,B1,B2,B3)
COMMON /SMBA/ XX,RXM
III=0
X=D3
Y=D2
Z=D1
CALL BASA(Z,Y,X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Z,Y,-X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Z,-Y,X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Z,-Y,-X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Y,X,Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Y,X,-Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Y,-X,Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Y,-X,-Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(X,Z,Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-X,Z,-Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-X,-Z,Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(X,-Z,-Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-X,-Y,-Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(X,-Y,Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(X,Y,-Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-X,Y,Z,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Z,-X,-Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Z,-X,Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Z,X,-Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Z,X,Y,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Y,-Z,-X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Y,-Z,X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(Y,Z,-X,III,TH,PR,B1,B2,B3)
IF(III.EQ.1) RETURN
CALL BASA(-Y,Z,X,III,TH,PR,B1,B2,B3)
RETURN
END

```



```

SUBROUTINE BASA(A1,A2,A3,III,TH,PR,B1,B2,B3)
COMMON /SMBA/ XX,RXM
IF(A3.LT.0..OR.A2.LT.0..OR.A1.LT.0.) RETURN
IF(A3.LT.A2.OR.A3.LT.A1) RETURN
B1=A1
B2=A2
B3=A3
III=1
IF(A3.GE.1.) THEN
TH=0.
PR=0.
RETURN
END IF
TH=ACOS(A3)/XX
PR=ATAN2(A2,A1)/XX
III=1
RETURN
END

```

```

SUBROUTINE AXIS(KR,RX,A1,A2,A3)
COMMON /SMA/ W(3,3)
R=2.*SIN(RX)
GO TO (1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,
*21,22,23,24),KR
1 A3=(W(1,2)-W(2,1))/R
  A2=(W(3,1)-W(1,3))/R
  A1=(W(2,3)-W(3,2))/R
  GO TO 25
2 A3=(-W(1,2)-W(2,1))/R
  A2=(-W(3,1)+W(1,3))/R
  A1=(W(2,3)+W(3,2))/R
  GO TO 25
3 A3=(-W(1,2)+W(2,1))/R
  A2=(W(3,1)+W(1,3))/R
  A1=(-W(2,3)-W(3,2))/R
  GO TO 25
4 A3=(W(1,2)+W(2,1))/R
  A2=(-W(3,1)-W(1,3))/R
  A1=(-W(2,3)+W(3,2))/R
  GO TO 25
5 A3=(W(2,2)-W(3,1))/R
  A2=(W(1,1)-W(2,3))/R
  A1=(W(3,3)-W(1,2))/R
  GO TO 25
6 A3=(-W(2,2)-W(3,1))/R
  A2=(-W(1,1)+W(2,3))/R
  A1=(W(3,3)+W(1,2))/R
  GO TO 25
7 A3=(-W(2,2)+W(3,1))/R
  A2=(W(1,1)+W(2,3))/R
  A1=(-W(3,3)-W(1,2))/R
  GO TO 25
8 A3=(W(2,2)+W(3,1))/R
  A2=(-W(1,1)-W(2,3))/R
  A1=(-W(3,3)+W(1,2))/R
  GO TO 25
9 A3=(W(3,2)-W(1,1))/R
  A2=(W(2,1)-W(3,3))/R
  A1=(W(1,3)-W(2,2))/R
  GO TO 25
10 A3=(-W(3,2)-W(1,1))/R
  A2=(-W(2,1)+W(3,3))/R
  A1=(W(1,3)+W(2,2))/R
  GO TO 25

```

```

11 A3=(-W(3,2)+W(1,1))/R
   A2=(W(2,1)+W(3,3))/R
   A1=(-W(1,3)-W(2,2))/R
   GO TO 25
12 A3=(W(3,2)+W(1,1))/R
   A2=(-W(2,1)-W(3,3))/R
   A1=(-W(1,3)+W(2,2))/R
   GO TO 25
13 A3=(-W(3,2)+W(2,1))/R
   A2=(-W(1,1)+W(3,3))/R
   A1=(-W(2,3)+W(1,2))/R
   GO TO 25
14 A3=(W(3,2)+W(2,1))/R
   A2=(W(1,1)-W(3,3))/R
   A1=(-W(2,3)-W(1,2))/R
   GO TO 25
15 A3=(W(3,2)-W(2,1))/R
   A2=(-W(1,1)-W(3,3))/R
   A1=(W(2,3)+W(1,2))/R
   GO TO 25
16 A3=(-W(3,2)-W(2,1))/R
   A2=(W(1,1)+W(3,3))/R
   A1=(W(2,3)-W(1,2))/R
   GO TO 25
17 A3=(-W(1,2)+W(3,1))/R
   A2=(-W(2,1)+W(1,3))/R
   A1=(-W(3,3)+W(2,2))/R
   GO TO 25
18 A3=(W(1,2)+W(3,1))/R
   A2=(W(2,1)-W(1,3))/R
   A1=(-W(3,3)-W(2,2))/R
   GO TO 25
19 A3=(W(1,2)-W(3,1))/R
   A2=(-W(2,1)-W(1,3))/R
   A3=(W(3,3)+W(2,2))/R
   GO TO 25
20 A3=(-W(1,2)-W(3,1))/R
   A2=(W(2,1)+W(1,3))/R
   A1=(W(3,3)-W(2,2))/R
   GO TO 25
21 A3=(-W(2,2)+W(1,1))/R
   A2=(-W(3,1)+W(2,3))/R
   A1=(-W(1,3)+W(3,2))/R
   GO TO 25
22 A3=(W(2,2)+W(1,1))/R
   A2=(W(3,1)-W(2,3))/R
   A1=(-W(1,3)-W(3,2))/R
   GO TO 25
23 A3=(W(2,2)-W(1,1))/R
   A2=(-W(3,1)-W(2,3))/R
   A1=(W(1,3)+W(3,2))/R
   GO TO 25
24 A3=(-W(2,2)-W(1,1))/R
   A2=(W(3,1)+W(2,3))/R
   A1=(W(1,3)-W(3,2))/R
25 CONTINUE
   RETURN
   END

```